# Symbolic software for separation of variables in the Hamilton-Jacobi equation for the L-systems. 

Yu.A. Grigoryev and A.V. Tsiganov<br>Department of Computational Physics, St.Petersburg State University, St.Petersburg, Russia<br>e-mail: tsiganov@mph.phys.spbu.ru


#### Abstract

We discuss computer implementation of the known algorithm of finding separation coordinates for the special class of orthogonal separable systems called L-systems or Benenti systems.


## 1 Introduction

Let $\mathcal{Q}$ be an-dimensional Riemannian manifold with generic local coordinates $q=\left(q^{1}, q^{2}, \ldots, q^{n}\right)$ and positive-definite metric tensor $\mathbf{G}$.

On the cotangent bundle $T^{*} \mathcal{Q}$ of $\mathcal{Q}$ with canonical coordinates $(p, q)$ we consider dynamical system with the natural Hamilton function

$$
\begin{equation*}
H=T(p, q)+V(q)=\sum_{i, j=1}^{n} \mathrm{~g}^{i j}(q) p_{i} p_{j}+V(q) \tag{1.1}
\end{equation*}
$$

Here $\mathrm{g}^{i j}(q)$ are components of the metric tensor $\mathbf{G}$ and $V(q)$ is the potential energy, a smooth function on $\mathcal{Q}$ canonically lifted to a function on $T^{*} \mathcal{Q}$.

One of the most effective ways of solving the corresponding equations of motion is by separation of variables in the Hamilton-Jacobi equation

$$
\begin{equation*}
H(p, q)=E . \tag{1.2}
\end{equation*}
$$

A coordinate system $Q=\left(Q^{1}, \ldots, Q^{n}\right)$ is called separable if the Hamilton-Jacobi equation admits a complete solution of the form

$$
\mathcal{S}(Q, \alpha)=\sum_{i=1}^{n} \mathcal{S}_{i}\left(Q^{i}, \alpha\right), \quad \operatorname{det}\left[\frac{\partial^{2} \mathcal{S}}{\partial Q^{i} \partial \alpha^{j}}\right] \neq 0
$$

Here $\alpha=\left(\alpha^{1}, \ldots, \alpha^{n}\right)$ is a set of separation constants. The corresponding Jacobi equations

$$
P_{i} \equiv \partial_{i} \mathcal{S}=\frac{\partial \mathcal{S}_{i}\left(Q^{i}, \alpha\right)}{\partial Q^{i}}
$$

are called the separated equations. In a similar manner, we will call a natural Hamiltonian or potential separable if such a separable coordinate system exists.

For a given Hamilton function $H(p, q)$ the problem of finding canonical transformation from initial variables $(p, q)$ to the separation coordinates $(P, Q)$ is very non-trivial. The problem was originally formulated by Jacobi when he invented elliptic coordinates and successfully applied them to solve several important mechanical problems, such as the problem of geodesic motion
on an ellipsoid and the Euler problem of planar motion in a force field of two attracting centers and the problem of the motion of three particles, which interact due to forces depending on their relative distances 1].

Up to now finding the separation coordinates for a given integrable system remains rather the magic art than a constructive theory. However, for the special class of the natural Hamiltonians we have a complete and algorithmic solution of this problem [4, 5, 6, 7]. The algorithm is straightforward enough to be implemented on a computer and thus turned into a practical tool.

In this note we present an implementation of this algorithm by means of the computer algebra system Maple 9.5. The corresponding Maple file with the code may be found in

```
http://www.maplesoft.com/applications/app_center_view.aspx?AID=1686
```


## 2 Algorithm of the point separation of variables

According to 4, 6] we restrict ourselves to the search of the point canonical transformations $Q=f(q)$ and $P=g(q, p)$ only.

In this case from Stäckel (1893), Levi-Civita (1904), Eisenhart (1934), Kalnins \& Miller (1980) and Benenti (1993) we have

Theorem 1 The Hamilton-Jacobi equation (1.2) is separable in orthogonal coordinates if and only if there exists symmetric Killing 2-tensor $\mathbf{K}$ with simple eigenvalues and normal eigenvectors such that

$$
\begin{equation*}
d(\mathbf{K} d V)=0 \tag{2.3}
\end{equation*}
$$

where d denotes the exterior derivative.
Such Killing tensor $\mathbf{K}$ is called characteristic, its existence is completely defined by the kinetic energy $T$. It means that the separation of the geodesic equation is a necessary condition for the separation of equation $T+V=E$. The equation $d(\mathbf{K} d V)=0$ is an integrability condition for the existence of the potential $V(q)$ that may be added to $T$.

In 1992 Benenti has shown a simple recurrence procedure to construct a special family of Killing tensors $\mathbf{K}$ obeying the assumptions of this theorem. He considered a special class of Riemannian manifolds $\mathcal{Q}$ endowed with the L-tensor, whose functionally independent eigenvalues are identified with the separation variables.

Following Benenti let us call L-tensor a conformal Killing tensor $\mathbf{L}$ with vanishing torsion and pointwise simple eigenvalues $Q_{i}$. Under these conditions the tensors

$$
\begin{equation*}
\mathbf{K}_{m}=\sum_{k=0}^{m} \sigma_{m-k} \mathbf{L}^{k}, \quad \text { or } \quad \mathbf{K}_{m}=\sigma_{m} \mathbf{G}-\mathbf{K}_{m-1} \mathbf{L}, \quad m=0, \ldots, n-1 \tag{2.4}
\end{equation*}
$$

where $\mathbf{L}$ is $(2,0)$ tensor field, are the Killing tensors with simple eigenvalues and normal eigenvectors. Here functions $\sigma_{m}$ are the elementary symmetric polynomials of degree $m$ on the eigenvalues of $\mathbf{L}$, such that $\operatorname{det}(\lambda I-\mathbf{L})=\sum_{m=0}^{n} \sigma_{m} \lambda^{n-m}$.

It can be shown [6] that equation $d(\mathbf{K} d V)=0$ implies $d\left(\mathbf{K}_{m} d V\right)=0$ for all the commuting Killing tensors. This allows to define the new potentials $V_{m}$ according to

$$
\begin{equation*}
d V_{m}=\mathbf{K}_{m} d V, \quad m=1, \ldots, n-1 \tag{2.5}
\end{equation*}
$$

and to introduce integrals of motion

$$
\begin{equation*}
H_{m}=\sum_{i, j=1}^{n} \mathbf{K}_{m}^{i j}(q) p_{i} p_{j}+V_{m}(q), \quad m=1, \ldots, n-1 \tag{2.6}
\end{equation*}
$$

These functions form a family of commuting integrals of the motion for the Hamiltonian $H$ (1.1).

According to 9 integrals of motion $H_{m}$ (2.6) are solutions of the recursion relations

$$
\begin{equation*}
d H_{m+1}=\mathbf{N}^{*} d H_{m}+\sigma_{m+1} d H, \quad m=1, \ldots, n-1, \quad H_{n} \equiv 0 \tag{2.7}
\end{equation*}
$$

Here $\mathbf{N}$ is the recursion operator, which is the complete lifting of $(1,1)$ tensor field $\mathbf{L}$ to $T^{*} \mathcal{Q}$

$$
\mathbf{N} \frac{\partial}{\partial q^{k}}=\sum_{i=1}^{n} L_{k}^{i} \frac{\partial}{\partial q^{i}}+\sum_{i j} p_{j}\left(\frac{\partial L_{i}^{j}}{\partial q^{k}}-\frac{\partial L_{k}^{j}}{\partial q^{i}}\right) \frac{\partial}{\partial p_{i}}, \quad \mathbf{N} \frac{\partial}{\partial p_{k}}=\sum_{i=1}^{n} L_{i}^{k} \frac{\partial}{\partial p_{i}}
$$

Recall, since a metric tensor is present, the boldface object $\mathbf{K}$ or $\mathbf{L}$ can be represented in components as a tensor of type $(2,0),(1,1)$ and $(0,2)$, respectively.

So, in order to get separation variables and integrals of motion for a given integrable system we have to find tensor $\mathbf{L}$ only. As noticed in [8], tensor $\mathbf{L}$ is an $L$-tensor with respect to the usual Riemannian metric G iff

$$
\begin{equation*}
d\left(\mathcal{L}_{X_{T}} \theta-T d \sigma_{1}\right)=0 \tag{2.8}
\end{equation*}
$$

where $\mathcal{L}$ is the Lie derivative along the geodesic vector field $X_{T}, \sigma_{1}=\operatorname{tr} \mathbf{L}$ is first symmetric polynomial and

$$
\begin{equation*}
\theta=\sum_{i, j=1}^{n} L_{j}^{i} p_{i} d q^{j} \tag{2.9}
\end{equation*}
$$

is the L-deformation of the standard Liouville 1-form $\theta_{0}=\sum p_{j} d q^{j}$ for any set of fibered coordinates $(p, q)$.

The equation $d(\mathbf{K} d V)=0$ with the Killing tensor $\mathbf{K}_{1}$ (2.4) may be rewritten in the similar form

$$
\begin{equation*}
d\left(\mathcal{L}_{X_{V}} \theta-V d \sigma_{1}\right)=0 \tag{2.10}
\end{equation*}
$$

see for instance 10.
Below we will use the following well-known expression for the Lie derivative $\mathcal{L}$ along the vector field $X$

$$
\mathcal{L}_{X}=\mathbf{i}_{X} d+d \mathbf{i}_{X}
$$

Here $i_{X}$ is a hook operator and $d$ is an exterior derivative. It allows us to decrease the amount of intermediate calculations as long as $d^{2}=0$ and we have

$$
d \mathcal{L}_{X}=d \dot{i}_{X} d+d^{2} \dot{i}_{X}=d \dot{i}_{X} d
$$

In these notations equations (2.8) and (2.10) read as

$$
\begin{align*}
d\left(\mathbf{i}_{X_{T}} d \theta-T d \sigma_{1}\right) & =0  \tag{2.11}\\
d\left(\mathbf{i}_{X_{V}} d \theta-V d \sigma_{1}\right) & =0 \tag{2.12}
\end{align*}
$$

We will call L-system or Benenti system any separable orthogonal system whose Killing tensor $\mathbf{K}$ in (2.3) is generated by an L-tensor according to (2.4). To construct the separation coordinates $Q$ for the L-system we have to solve equations (2.11) and (2.12) with respect to functions $L_{j}^{i}(q)$ and to find the eigenvalues of the tensor $\mathbf{L}$.

In the next section we present a computer program for search of the L-tensors and the associated separation coordinates for L-systems.

Remark: The 1-form $\theta$ may be used to construct the second Poisson structure on $T^{*} \mathcal{Q}$, 10 . In this case the recursion operator $N$ will be a complete lifting of the L-tensor $\mathbf{L}$ from the configuration space to the whole phase space. Therefore, the coordinate separation of variables can be treated as a particular case of the bi-hamiltonian theory of separation of variables.

## 3 The program for the search of separation variables

In this section we present an implementation of the discussed algorithm made in the symbolic computational system Maple v.9.5.

We will use the standard Maple package liesymm, which was designed for construction of the differential forms corresponding to partial differential equations. In fact we are doing the inverse procedure, i.e. starting with the differential forms (2.112.12) we have to get a system of partial differential equations.

Let us start with the following command

```
> with(liesymm):
```

which makes the short form names of the functions of a Maple package available at the interactive level.

At the first step we need to determine dimension of a given configuration space $\mathcal{Q}$ and to suppose that the phase space $T^{*} \mathcal{Q}$ is equipped with some canonical coordinates $q=\left(q^{1}, \ldots, q^{n}\right)$ and $p=\left(p_{1}, \ldots, p_{n}\right)$ :
$>\mathrm{n}:=2$;

```
            n:=2
> q:=seq(q||i,i=1..n): p:=seq(p||i,i=1..n): var:=q,p:
> setup(var);
```

$$
[q 1, q 2, p 1, p 2]
$$

Now we will look for a tensor field $\mathbf{L}$ with vanishing torsion and functionally independent eigenvalues. For a given integrable system tensor $L$ has to satisfy equations (2.11) and (2.12) which include first symmetric polynomial $\sigma_{1}$ on the eigenvalues of $\mathbf{L}$

```
> sigma:=add(L[i,i] (q),i=1..n);
    \(\sigma:=L_{1,1}(q 1, q 2)+L_{2,2}(q 1, q 2)\)
```

and L-deformation $\theta(2.9)$ of the standard Liouville form

```
> theta:=add(add(L[i,j](q)*p||i*d(q||j),i=1..n),j=1..n);
0:= L\mp@code{1,1 (q1,q2) p1d(q1)+ L2,1 (q1, q2) p2 d(q1)}
    + L L,2(q1,q2) p1d(q2)+ L L2,2 (q1,q2) p2 d(q2)
```

Here components of L-tensor $L_{j}^{i}(q)$ are the unknown functions on the configuration space $\mathcal{Q}$. This L-tensor gives rise to a torsionless tensor field $\mathbf{N}$ of type $(1,1)$ on $T^{*} \mathrm{Q}$, which acts on the 1-form of fibered coordinates $(p, q)$ as

```
> Nq:={seq(d(q[k])=add(L[k,j] (q)*d(q[j]) ,j=1..n),k=1..n)}:
> Np:={seq(d(p[k])=add(L[j,k](q)*d(p[j])
> - pl|j*add((diff(L[j,i](q),q[k]) - diff(L[j,k](q),q[i]))*d(q||i),i=1..n),
> j=1..n),k=1..n)}:
```

At the second step we introduce canonical Poisson tensor $\mathscr{P}$

```
> ed:=array(identity, 1..n,1..n): u:=array(sparse,1..n,1..n):
> P:=linalg[stackmatrix] (linalg[augment](u,ed),
    linalg[augment](-ed,u));
        P:=[}[\begin{array}{rrrrr}{0}&{0}&{1}&{0}\\{0}&{0}&{0}&{1}\\{-1}&{0}&{0}&{0}\\{0}&{-1}&{0}&{0}\end{array}
```

and canonical Poisson brackets

```
> PB:=proc(f,g) options operator, arrow:
> add( diff(f,p||i)*diff(g,q||i)- diff(f,q||i)*diff(g,p||i), i=1..n)
> end:
```

It allows us to calculate the vector fields $X_{T}=\mathscr{P} d T(p, q)$ and $X_{V}=\mathscr{P} d V(q)$. They are equal to
$>\mathrm{dT}:=\operatorname{array}(1 . .2 * \mathrm{n})$ :
$>$ for $i$ from 1 to $2 * n$ do dT[i]:=diff(T(var), var[i]): end do:
> X_T:=evalm(P\&*dT);

$$
\begin{aligned}
X T:= & {\left[\frac{\partial}{\partial p 1} \mathrm{~T}\left(q 1, q^{2}, p 1, p 2\right), \frac{\partial}{\partial p^{2}} \mathrm{~T}\left(q 1, q^{2}, p 1, p 2\right)\right.} \\
& \left.-\left(\frac{\partial}{\partial q 1} \mathrm{~T}\left(q 1, q^{2}, p 1, p 2\right)\right),-\left(\frac{\partial}{\partial q^{2}} \mathrm{~T}\left(q 1, q^{2}, p 1, p 2\right)\right)\right]
\end{aligned}
$$

and
$>d V:=\operatorname{array}(1 . .2 * n)$ :
$>$ for $i$ from 1 to $2 * n$ do $d V[i]:=\operatorname{diff}(V(q), \operatorname{var}[i])$ : end do:
> X_V:=evalm(P\&*dV);

$$
X V:=\left[0,0,-\left(\frac{\partial}{\partial q 1} \mathrm{~V}(q 1, q 2)\right),-\left(\frac{\partial}{\partial q 2} \mathrm{~V}(q 1, q 2)\right)\right]
$$

The equations with the vector field $X_{V}$ will be simpler due to its zero components than ones with $X_{T}$ and, therefore, we start with the second equation (2.12) in order to explain some features of the Maple procedures. Namely, substituting the vector field $X_{V}$ and 1-form $\theta$ into the $i_{X_{V}} d \theta$, one gets

```
> idT:=wcollect(hook(d(theta),X_V)):
```

$$
\begin{aligned}
i d T:=( & -L_{2,1}(q 1, q 2)\left(\frac{\partial}{\partial q 2} \mathrm{~V}(q 1, q 2)\right)(q 1, q 2, p 1, p 2) \\
& \left.-L_{1,1}(q 1, q 2)\left(\frac{\partial}{\partial q 1} \mathrm{~V}(q 1, q 2)\right)(q 1, q 2, p 1, p 2)\right) \mathrm{d}(q 1) \\
+( & -L_{2,2}(q 1, q 2)\left(\frac{\partial}{\partial q 2} \mathrm{~V}(q 1, q 2)\right)(q 1, q 2, p 1, p 2) \\
& \left.-L_{1,2}(q 1, q 2)\left(\frac{\partial}{\partial q 1} \mathrm{~V}(q 1, q 2)\right)(q 1, q 2, p 1, p 2)\right) \mathrm{d}(q 2)
\end{aligned}
$$

It is easy to see that this expression contains derivatives of the potential $V(q)$, which formally depend on all the variables $(q, p)$. In fact these derivatives depend on the variables $q$ only. It is an unpleasant feature of the Maple procedure hook from the package liesymm.

In order to get the correct expression for $i_{X_{V}} d \theta$ we have to use the special substitution

```
> Trans:={seq(diff(V(q),q||i)(var)=diff(V(q),q||i),i=1..n)}:
> idT:=subs(Trans,idT);
```

$$
\begin{aligned}
i d T: & =\left(-L_{2,1}(q 1, q 2) \frac{\partial}{\partial q 2} \mathrm{~V}(q 1, q 2)-L_{1,1}(q 1, q 2) \frac{\partial}{\partial q 1} \mathrm{~V}(q 1, q 2)\right) \mathrm{d}(q 1) \\
& +\left(-L_{2,2}(q 1, q 2) \frac{\partial}{\partial q 2} \mathrm{~V}(q 1, q 2)-L_{1,2}(q 1, q 2) \frac{\partial}{\partial q 1} \mathrm{~V}(q 1, q 2)\right) \mathrm{d}(q 2)
\end{aligned}
$$

The equation (2.12) are satisfied identically and coefficients for independent 2-forms in (2.12) vanish. It gives rise to the following system of equations

```
> SysEq:=annul(d(idT-V(q)*d(sigma)), [var] ): nops(SysEq);
```

At $n=2$ equation (2.12) generates only one partial differential equation, while at $n=3$ the system SysEq consists of three equations and so on.

We keep these equations in a special list

```
> ListEq:=NULL:
> for i from 1 to nops(SysEq) do
> ListEq:=ListEq,lhs(SysEq[i]);
> end do:
```

Now we pass on the remaining equation (2.11). As above we have to simplify this equation through the additional substitution for derivatives after the application of the hook operator

```
> Eq:=wcollect(d(hook(d(theta),X_T)-T(var)*d(sigma))):
> Trans:={seq( diff(T(var),var[i])(var)=diff(T(var),var[i]),i=1..2*n)}:
```

Expanding equation (2.11) in the basis of 2 -forms one gets one more system of equations

```
> SysEq:=annul(subs(Trans,Eq),[var]): nops(SysEq);
```

At $n=2,3$ equation (2.12) gives rise to six and fifteen equations respectively.
By adding these equations to earlier prepared list of equations ListEq one gets a complete set of algebraic and partial differential equations on the components of L-tensor.

```
> for i from 1 to nops(SysEq) do ListEq:=ListEq,lhs(SysEq[i]); end do:
> NumEq:=nops([ListEq]);
```

$$
N u m E q:=7
$$

At $n=2,3$ one gets seven and eighteen equations on four and nine functions $L_{j}^{i}(q)$ respectively.
At the third step we have to define the Hamilton function. As an example we consider the Hénon-Heiles integrable model with the Hamiltonian

$$
\begin{equation*}
H=p_{1}^{2}+p_{2}^{2}+\frac{1}{2}\left(q_{1}^{2}+q_{2}^{2}\right)+q_{1}^{2} q_{2}+2 q_{2}^{3} \tag{3.13}
\end{equation*}
$$

In this case kinetic energy $T$ and potential $V(q)$ are given by

```
> Tn:=add(p||i^2,i=1..n);
> Vn:=1/2*(q1^2+q2^2) +q1^2*q2+2*q2^3;
```

Substituting these parts of the Hamilton function into the prepared above list of equations ListEq one gets a set of polynomial equations of the second degree in the momenta $p$, which must be identically satisfied for all admissible values of variables $p$. It means that coefficients for the second, first and zeroth power of $p_{i}$ vanish. All these coefficients form new system of algebraic and partial differential equations on the functions $L_{j}^{i}(q)$.

```
> System:= NULL:
> for i from 1 to NumEq do
> u:=expand(dvalue(subs({ T(var)=Tn, V (z)=Vn },ListEq[i]))):
> System:=System,coeffs(u,{p}):
> end do:
> NumSys:=nops({System});
```

$$
\text { NumSys }:=13
$$

At $n=2,3,4$ the resulting system of equation consists of 13,51 and 136 equations. Even though some of these equations may be dependent, it is convenient to use them all simultaneously. Of course, for a generic potential this system is heavily overdetermined and has only the trivial solution, which means that the Hamilton function is non-separable.

In the case of polynomial potentials we can easy solve this system by means of the standard Maple procedure pdsolve

```
> Ans:=pdsolve( {System}, {seq(seq(L[i,j](z), i=1..n),j=1..n)} );
```

which allows us to get the following answer

$$
\begin{aligned}
\text { Ans }:= & \left\{L_{1,1}(q 1, q \mathcal{Z})=\frac{3 C_{1}}{4}+C_{2}, L_{1,2}(q 1, q \mathcal{2})=\frac{C_{1} q 1}{2}\right. \\
& \left.L_{2,2}(q 1, q \mathcal{2})=C_{1} q \mathcal{2}+C_{2}, L_{2,1}(q 1, q \mathcal{2})=\frac{C_{1} q 1}{2}\right\}
\end{aligned}
$$

depending on two arbitrary constants $C_{1,2}$.
Substituting this answer into the L-tensor one gets

```
> Trans:={seq(seq(L[i,j] (q)=L[i,j],i=1..n),j=1..n)}:
> Ans:=simplify(map2(subs,Trans,Ans)):
> L:=array(1..n,1..n):
> L:=map2(subs,Ans,evalm(L));
\[
L:=\left[\begin{array}{cc}
\frac{3}{4} C_{1}+C_{2} & \frac{C_{1} q 1}{2} \\
\frac{C_{1} q 1}{2} & C 1 q 2+C_{2}
\end{array}\right]
\]
```

The eigenvalues of this L-tensor are the separation coordinates

```
> Q:=linalg[eigenvalues] (Ln);
```

$$
\begin{aligned}
Q:= & \frac{3 C_{1}}{8}+C_{2}+\frac{C_{1} q^{2}}{2}+\frac{\sqrt{9 C_{1}^{2}-24 C_{1}^{2} q \mathcal{2}+16 C_{1}^{2} q \mathcal{Q}^{2}+16 C_{1}^{2} q 1^{2}}}{8} \\
& \frac{3 C_{1}}{8}+C_{2}+\frac{C_{1} q \mathcal{2}}{2}-\frac{\sqrt{9 C_{1}^{2}-24 C_{1}^{2} q \mathcal{2}+16 C_{1}^{2} q \mathcal{D}^{2}+16 C_{1}^{2} q 1^{2}}}{8} .
\end{aligned}
$$

Thus one gets translated parabolic coordinates, which are the separation coordinates for the Hénon-Heiles model [7]. It takes less then a minute of the computer time.

More explicitly, we obtain a family of equivalent separable coordinate systems labeled by $C_{1,2}$. Recall, that two separable systems are called equivalent if the corresponding separated solutions of the Hamilton-Jacobi equation generate the same Lagrangian foliation of $T^{*} \mathcal{Q}$.

Now let us consider construction of the corresponding integrals of motion in the framework of the Riemannian geometry. Following Benenti we have to construct polynomials $\sigma_{m}$

```
>for i from 0 to n do
> sigma[i]:=coeff(linalg[det](lambda*ed-L),lambda,n-i):
> end do:
```

and basis of Killing tensors $\mathbf{K}_{m}$ (2.4)

```
> for m from 1 to n-1 do
> K||m:=evalm(add( sigma[m-k]*L^k,k=0..m)):
> end do:
```

Remind, that in our case $\mathrm{g}^{i j}=1$ and, therefore, $L^{i j}=L_{j}^{i}$.
In addition to $\mathbf{K}_{m}$ equations (2.5) consist of exterior derivatives on the unknown potentials $d V_{m}$, which are defined by

```
> for m from 0 to n-1 do
> dV||m:=array(1..n): for i from 1 to n do
> dV||m[i]:=diff(V||m(q),q[i]):
end do:
> end do:
> dV0:=map2(subs,VO(q)=Vn,dV0):
```

Now we can solve equations (2.5) and determine integrals of motion $H_{m}$ (2.6)

```
for m from 1 to n-1 do
> eqV:=evalm(dV||m-K||m&*dV0):
> Ans:=pdsolve({seq(eqV[i], i=1..n)},V||m(q)):
> H[m]:=add(add( K||m[i,j]*p||i*p||j,i=1..n),j=1..n)+subs(Ans,V||m(q)):
> end do:
```

We can show the Hamilton function

```
> H[0]:=Tn+Vn;
```

$$
H_{0}:=p 1^{2}+p \mathcal{2}^{2}+\frac{q 1^{2}}{2}+\frac{q \mathscr{2}^{2}}{2}+q 1^{2} q \mathscr{2}+2 q \mathcal{2}^{3}
$$

and prove that our integrals of motion $H_{m}$ commute with the Hamiltonian $H_{0}$

```
for m from 1 to n-1 do
Z ZERO:=simplify(PB(H[0],H[m]));
> end do;
```

$$
Z E R O:=0
$$

Remark: At $n \leq 10$ and for the polynomial potentials we can solve overdetermined system of equation System with the standard procedure pdsolve on a standard personal computer in the reasonable time. At $n>10$ we have to use special computers or special symbolic software for solving heavily overdetermined systems of equations.

## 4 Examples

In this section we present some examples which illustrate various aspects of the proposed code.

### 4.1 The anharmonic oscillator

Configuration space $\mathcal{Q}$ is the $n$-dimensional Euclidean space $\mathbb{R}^{n}$ with cartesian coordinates $q$ such that metric takes the form $d s^{2}=\sum \mathrm{g}_{i j} d q_{i} d q_{j}=\sum d q_{i}{ }^{2}$. Let us consider the anharmonic oscillator with the Hamiltonian

$$
H=\sum_{i=1}^{n} p_{i}^{2}+\sum_{i=1}^{n} a_{i} q_{i}^{2}+\left(\sum_{i=1}^{n} q_{i}^{2}\right)^{2}
$$

We have to substitute this Hamiltonian instead of (3.13) using commands

```
> Tn:=add(p||i^2,i=1..n);
> Vn:=add(a| |i*q| |i^2,i=1..n)+(add(q| |i^2,i=1..n))^2;
```

In is easy to calculate solution of the complete system of equations System obtained from equations (2.11) and (2.12)

$$
L_{j}^{i}(q)=\left(C_{1}+a_{i}-a_{n}\right) \delta_{i j}+C_{2} q_{i} q_{j}
$$

Here the solution found by Maple is presented without any further simplification.
At $C_{1}=a_{n}$ and $C_{2}=1$ the eigenvalues $Q_{i}$ of the corresponding tensor $\mathbf{L}$ are defined by the equation

$$
\begin{equation*}
\frac{\operatorname{det}(L-\lambda)}{\prod_{i=1}^{n}\left(\lambda-a_{i}\right)} \equiv-1+\sum_{i=1}^{n} \frac{q_{i}^{2}}{\lambda-a_{i}}=\prod_{i=1}^{n} \frac{\lambda-Q_{i}}{\lambda-a_{i}} \tag{4.14}
\end{equation*}
$$

This is the well-known relation which defines standard elliptic coordinates in $\mathbb{R}^{n}$.

These coordinates were introduced by Jacobi in a note in Crelle's Journal 11. A thorough discussion of its general properties as well as of its use for separation of variables in the HamiltonJacobi equation can be found in his lecture notes 1].

Now let us consider construction of integrals of motion in the involution in framework of the bi-Hamiltonian geometry. At first we determine action of the recursion operator $\mathbf{N}$ on the 1 -forms $d p$ and $d q$ using given tensor $\mathbf{L}$

```
> Nqn:=wcollect(simplify(map2(subs,Ans,Nq))):
> Npn:=wcollect(simplify(map2(subs,Ans,Np))):
```

Then we calculate symmetric polynomials $\sigma_{m}$ on eigenvalues of $\mathbf{L}$

```
> for m from 0 to n do
> sigma[m]:=coeff(linalg[det](Ln-lambda*ed),lambda,n-m):
> end do:
```

At the final step we construct and solve recursion relations (2.7)

```
> unassign(H): H[n+1]:=0: H[0]:=Tn+Vn;
> for m from n-1 by -1 to 1 do
> Eq:=wcollect(dvalue(d(H[m+1])-
> subs(Nqn union Npn,d(H[m](var)))-sigma[m+1]*d(H[0]))):
> Sys:=annul(Eq,[var]):
> Ans1:=pdsolve(Sys,H[m](var));
> H[m]:=subs(Ans1,H[m](var)):
> end do:
```

In addition we can check the Poisson brackets relations between integrals of motion

```
> for m from 1 to n-1 do
> ZERO:=simplify(PB(H[0],H[m]));
> end do;
```


### 4.2 The Euler two centers problem

Let us consider the Euler problem of planar motion in a force field of two attracting centers with the Hamiltonian

$$
H=p_{1}^{2}+p_{2}^{2}-\left(\frac{a_{1}}{\sqrt{\left(q_{1}-c\right)^{2}+q_{2}^{2}}}+\frac{a_{2}}{\sqrt{\left(q_{1}+c\right)^{2}+q_{2}^{2}}}\right)
$$

In contrast with the previous examples potential $V(q)$ is an algebraic function instead of a polynomial one.

Solution of the overdetermined system of equations obtained from (2.11) and (2.12) is equal to

$$
L=\left[\begin{array}{cc}
\left(q_{1}^{2}-c^{2}\right) C_{2}+C_{1} & q_{1} q_{2} C_{2} \\
q_{1} q_{2} C_{2} & q_{2}^{2} C_{2}+C_{1}
\end{array}\right]
$$

At $C_{1}=1 / 2 c^{2}$ and $C_{2}=1$ the eigenvalues of the matrix $L$ are the standard elliptic coordinates

$$
1-\frac{q_{1}^{2}}{\lambda+c^{2} / 2}-\frac{q_{2}^{2}}{\lambda-c^{2} / 2}=\frac{\left(\lambda-Q_{1}\right)\left(\lambda-Q_{2}\right)}{\left(\lambda-c^{2} / 2\right)\left(\lambda+c^{2} / 2\right)}
$$

### 4.3 The Toda lattice

As above, configuration space $\mathcal{Q}$ is the $n$-dimensional Euclidean space $\mathbb{R}^{n}$ with cartesian coordinates $q$. The Hamiltonian of the periodic Toda lattice is given by

$$
\begin{equation*}
H=\frac{1}{2} \sum_{i=1}^{n} p_{i}^{2}+\sum_{i=1}^{n} \exp \left(q_{i}-q_{i+1}\right), \quad q_{n+1} \equiv q_{1} \tag{4.15}
\end{equation*}
$$

At $n=2$ there exists the unique solution of the complete system of equations System

$$
L=\left[\begin{array}{ll}
C_{1} & C_{2} \\
C_{2} & C_{1}
\end{array}\right]
$$

Of course, eigenvalues of this matrix are not the separation variables. However, in this case the number of free parameters is equal to the dimension of the Riemannian manifold $\mathcal{Q}$. It allows us to construct the separation coordinates too (see [7]). Namely, let us diagonalize matrix $L$

$$
L=V^{-1} \operatorname{diag}\left(C_{1}+C_{2}, C_{1}-C_{1}\right) V, \quad V=\left(\begin{array}{cc}
1 & 1 \\
-1 & 1
\end{array}\right)
$$

It is easy to see that the separation coordinates $Q$ are the following cartesian coordinates

$$
Q=V q, \quad \Rightarrow \quad Q_{1}=q_{1}+q_{2}, \quad Q_{2}=q_{1}+q_{2}
$$

At $n=3$ solution of the equations System is the trivial o

$$
L=\left[\begin{array}{lll}
C_{1} & C_{2} & C_{2} \\
C_{2} & C_{1} & C_{2} \\
C_{2} & C_{2} & C_{1}
\end{array}\right]
$$

The number of free parameters is less than the dimension of the Riemannian manifold. It means that this Hamiltonian does not separable through point transformations. However, we have to underline that this Hamiltonian admits separation of variables in a wider class of canonical transformations of the whole phase space.

### 4.4 The Neumann system

The unit sphere $\mathbb{S}^{n}$ is the Riemannian subspace of $\mathbb{R}^{n+1}$ whose points have cartesian coordinates $x=\left(x_{1}, \ldots, x_{n+1}\right)$ satisfying $|x|=\sqrt{(x, x)}=1$.

The Neumann system is a well-known and very much studied mechanical system, which describes particle moving on the sphere under the influence of a quadratic potential $V=$ $\frac{1}{2} \sum a_{i} x_{i}^{2}$. Due to the form of the constraint and of the potential, it is quite natural to perform all the computations in the local coordinates $q_{i}=x_{i}^{2}, i=1, \ldots, n$. In these coordinates the Hamiltonian of the Neumann system is given by $H=T+V$ where

$$
T=2 \sum_{i=1}^{n} q_{i}\left(1-q_{i}\right) p_{i}^{2}-4 \sum_{i<j}^{n} q_{i} q_{j} p_{i} p_{j}, \quad V=\frac{1}{2} \sum_{i=1}^{n}\left(a_{i}-a_{n+1}\right) q_{i}
$$

Here $p_{i}$ are momenta conjugated to $q_{i}$ and $a_{1}, \ldots, a_{n+1}$ are arbitrary parameters [10]. We can insert this Hamiltonian in the Maple code with the commands

```
> Tn:=2*add(q||i*(1-q||i)*p||i^2,i=1..n)
> -4*add(add(q||i*q||*p||i*p||j,j=i+1..n),i=1..n);
> Vn:=1/2*add((a||i-a||(N+1))*q||i,i=1..n);
```

At $n=2$ the unique solution of the complete system of equations System looks like

$$
L=\left[\begin{array}{cc}
C_{2}-\frac{q_{1}\left(a_{3}-a_{1}\right)+a_{1}-a_{2}}{a_{2}-a_{3}} C_{1} & q_{1} C_{1} \\
\frac{q_{2}\left(a_{1}-a_{3}\right)}{a_{2}-a_{3}} C_{1} & q_{2} C_{1}+C_{2}
\end{array}\right]
$$

Recall, that $L_{j}^{i}(q)$ are components of the tensor $\mathbf{L}$, which is symmetric with respect to the metric G. For the non-flat metrics the $L$-matrices obtained in Maple are non-symmetric.

If $C_{1}=a_{3}-a_{2}$ and $C_{2}=a_{2}$ the eigenvalues $Q_{1,2}$ of this L-tensor are zeroes of the following function

$$
e(\lambda)=\frac{\operatorname{det}(L-\lambda)}{\prod_{i=1}^{n}\left(\lambda-a_{i}\right)}=\frac{q_{1}}{\lambda-a_{1}}+\frac{q_{2}}{\lambda-a_{2}}+\frac{1-q_{1}-q_{2}}{\lambda-a_{3}}
$$

In the redundant coordinates $x_{i}$ one gets the following definition of the separation coordinates

$$
\sum_{j=1}^{n+1} \frac{x_{j}^{2}}{\lambda-a_{j}}=\frac{\prod_{i=1}^{n}\left(\lambda-Q_{i}\right)}{\prod_{j=1}^{n+1}\left(\lambda-a_{j}\right)}, \quad \text { with } \quad \sum_{i=1}^{n+1} x_{j}^{2}=1
$$

These are well-known defining relations for the spheroconical or elliptic-spherical coordinates. At $n>2$ one gets the same relations too.

In contrast with the previous examples in this case wehave non-trivial metric and $L_{j}^{i}(q) \neq$ $L^{i j}(q)$. In order to get integrals of motion it is more convenient to use recursion relations (2.7). The corresponding part of code looks like

```
> Nqn:=wcollect(simplify(map2(subs,Ans,Nq))):
> Npn:=wcollect(simplify(map2(subs,Ans,Np))):
> for m from 0 to n do
> sigma[m]:=coeff(linalg[det](Ln-lambda*ed),lambda,n-m):
> end do:
> unassign(H): H[n+1]:=0: H[0]:=Tn+Vn;
>for m from n-1 by -1 to 1 do
 Eq:=wcollect(dvalue(d(H[m+1])-
> subs(Nqn union Npn,d(H[m](var)))-sigma[m+1]*d(H[0]))):
> Sys:=annul(Eq,[var]):
> Ans1:=pdsolve(Sys,H[m](var));
> H[m]:=subs(Ans1,H[m](var)):
> end do:
```

Of course, we can check the Poisson brackets relations between these integrals of motion

```
> for m from 1 to n-1 do
Z ZERO:=simplify(PB(H[0],H[m]));
> end do;
```

At $n=2,3$ all the calculations take about one minute and eight minutes respectively.
Remark: In contrast with [10] we directly solve equations (2.8 2.10) without any additional assumptions about the affine structure of the solutions. It allows us to prove that there is one unique solution only.

### 4.5 The Jacobi-Calogero inverse-square model

Configuration space $\mathcal{Q}$ is 3 -dimensional Euclidean space $\mathbb{R}^{3}$ with cartesian coordinates $q=$ $\left(q_{1}, q_{2}, q_{3}\right)$. The Hamilton function is given by

$$
H=\sum_{i=1}^{3} p_{i}^{2}+\left(q_{1}-q_{2}\right)^{-2}+\left(q_{2}-q_{3}\right)^{-2}+\left(q_{3}-q_{1}\right)^{-2}
$$

For this Hamiltonian solution of the complete system of equations System depends on four arbitrary parameters

$$
L=\left[\begin{array}{ccc}
q_{1}^{2} C_{1}+2 q_{1} C_{2}+C_{3} & q_{1} q_{2} C_{1}+\left(q_{1}+q_{2}\right) C_{2}+C_{4} & q_{1} q_{3} C_{1}+\left(q_{1}+q_{3}\right) C_{2}+C_{4} \\
q_{1} q_{2} C_{1}+\left(q_{1}+q_{2}\right) C_{2}+C_{4} & q_{2}^{2} C_{1}+2 q_{2} C_{2}+C_{3} & q_{2} q_{3} C_{1}+\left(q_{2}+q_{3}\right) C_{2}+C_{4} \\
q_{1} q_{3} C_{1}+\left(q_{1}+q_{3}\right) C_{2}+C_{4} & q_{2} q_{3} C_{1}+\left(q_{2}+q_{3}\right) C_{2}+C_{4} & q_{3}^{2} C_{1}+2 q_{3} C_{2}+C_{3}
\end{array}\right]
$$

The eigenvalues of this L-tensor are the separation coordinates

```
> Q:=linalg[eigenvalues] (L);
```

The number of free parameters $C_{k}$ is more than the dimension of the Riemannian manifold. It means that this integrable system is degenerate or superintegrable system, which admits separation of variables in some different coordinate systems. Namely, at the different values of the parameters $C_{1}, \ldots, C_{4}$ eigenvalues of the tensor $\mathbf{L}$ are oblate spheroidal, prolate spheroidal, spherical, rotational parabolic and circular cylindrical coordinates. The detailed discussion may be found in [7, 12].

### 4.6 An elliptic egg

Let us consider the Hamiltonian with the rational potential

$$
H=\sum_{i=1}^{n} p_{i}^{2}+\frac{c}{\left(1-\sum_{i=1}^{n} \frac{q_{i}^{2}}{a_{i}^{2}}\right)}, \quad c, a_{i} \in \mathbb{R}
$$

In contrast with the previous examples potential $V(q)$ is a rational function rather than a polynomial one. This potential has a singular surface, which is an ellipsoid

$$
\sum_{i=1}^{n} \frac{q_{i}^{2}}{a_{i}^{2}}=1
$$

For $c>0$, this singular surface is repelling inwards, so that all trajectories starting inside the ellipsoid remain there forever.

At $n>2$ the Maple procedure pdsolve can not solve the system of equations System obtained from the both equations (2.11) and (2.12) in the reasonable time.

In this case we have to solve equation for the geodesic motion (2.11) and then the equation on potential (2.12). At $n \leq 10$ it takes just a few minutes.

In our case solution of the system of equations generated by the geodesic equation (2.11) is equal to

$$
L_{j}^{i}(q)=\alpha q_{i} q_{j}+A_{i j} q_{i}+B_{i j} q_{j}+C_{i j}
$$

Here coefficients $\alpha, A_{i j}, B_{i j}$ and $C_{i j}$ are arbitrary constants, $A_{i j}=B_{j i}$ and $C_{i j}=C_{j i}$.
Substituting this solution into the system of equations obtained from (2.12) one gets a system of algebraic equations on the coefficients $A_{i j}, B_{i j}$ and $C_{i j}$. The standard Maple program solve easy solves this system of algebraic equations in few seconds. The eigenvalues of the corresponding tensor $\mathbf{L}$ are the standard elliptic coordinates 4.14).

## 5 Conclusion

We present the first part of the symbolic software which builds the separation coordinates in the Hamilton-Jacobi equation for the L-systems. The second part of this software is devoted to building and solving the corresponding separated equations. We leave this part to subsequent publications.

Recall, that Maple file with the first part of the code may be found in the following URL

```
http://www.maplesoft.com/applications/app_center_view.aspx?AID=1686
```

The latest version of the Maple file may be obtained from authors.

## References

[1] Jacobi C.G.J., Vorlesungen über dynamik, Georg Reimer, Berlin, 1866, Jacobi's lectures on dynamics given in Königsberg 1842-1843, published by A.Clebsch.
[2] Stäckel P. Uber die Integration der Hamilton-Jacobischen Differential Gleichung Mittelst Separation der Variabel, Habilitationsschrift, Halle, 1891.
[3] Levi-Civita T., Integrazione delle equazione di Hamilton-Jacobi per separazione di variabili, Math. Ann., v.24, p.383-397, 1904.
[4] Eisenhart L.P., Separable systems of Stäckel, Ann.Math., v.35, p.284-305, 1934.
[5] Kalnins E.G., W. Miller, Jr., Killing tensors and variable separation for Hamilton-Jacobi and Helmholtz equations, SIAM J. Math. Anal., v.11, p.1011-1026, 1980.
[6] Benenti, S. Orthogonal separable dynamical systems. In Proceedings of the 5th International Conference on Differential Geometry and Its Applications, Silesian University at Opava, August 24-28, 1992, O.Kowalski and D.Krupka Eds. Differential Geometry and Its Applications, v.1, p.163-184, 1993.
[7] Rauch-Wojciechowski S,, Waksjö C., How to find separation coordinates for the HamiltonJacobi equation: a criterion of separability for natural Hamiltonian systems, Math. Phys. Anal. Geom., v.6, no.4, p.301-348, 2003.
[8] Crampin M., Sarlet W., Thompson G., Bi-differential calculi, bi-Hamiltonian systems and conformal Killing tensors. J. Phys. A, v.33, p.8755-8770, 2000.
Crampin M., Conformal Killing tensors with vanishing torsion and the separation of variables in the Hamilton-Jacobi equation, Diff. Geom. Appl., v.18, p.87-102, 2003.
[9] Ibort A., Magri F., Marmo G., Bihamiltonian structures and Stäckel separability, J. Geometry and Physics, v.33, p.210-228, 2000.
[10] Bartocci C., Falqui G., Pedroni M., A geometric approach to the separability of the Neumann-Rosochatius system, Differential Geom. Appl., v.21, no.3, p.349-360, 2004.
[11] Jacobi C. G. J., J. Reine Angew. Math., XIX.
[12] Benenti S., Chanu C., Rastelli G., The super-separability of the three-body inverse-square Calogero system, J. Math. Phys. v.41, p.4654-4678, 2000.

